```
> restart:with(LinearAlgebra):
>
> # General sphere condition
> Lambda:=-R^2*e0^2+b^2*e0^2+c^2*e0^2+A^2*e0^2+B^2*e0^2+C^2*e0^2+a
  ^2*e0^2-R^2*e1^2+b^2*e1^2+c^2*e1^2+A^2*e1^2+B^2*e1^2+C^2*e1^2+a^
  2*e1^2-R^2*e2^2+b^2*e2^2+c^2*e2^2+A^2*e2^2+B^2*e2^2+C^2*e2^2+a^2
  *e2^2-R^2*e3^2+a^2*e3^2+b^2*e3^2+c^2*e3^2+A^2*e3^2+B^2*e3^2+C^2*
  e3^2+4*A*f0*e1+4*C*f1*e2-2*A*a*e0^2+2*A*a*e3^2-4*c*e3*f0-4*b*e3*
  f1+4*a*e3*f2+4*a*e0*f1+4*c*e0*f3+4*A*f2*e3-4*A*f1*e0-4*C*f2*e1+4
  *b*e0*f2-2*C*c*e0^2+2*C*c*e2^2+4*C*f0*e3-4*b*e2*f0-4*a*e2*f3+2*C
  *c*e1^2+4*B*f3*e1-2*B*b*e2^2-4*B*f1*e3+4*B*f0*e2-4*c*e1*f2-2*A*a
  *e1^2-4*A*f3*e2+2*B*b*e3^2-4*B*f2*e0-2*B*b*e0^2-4*a*e1*f0-2*C*c*
  e3^2-4*C*f3*e0+4*c*e2*f1+2*B*b*e1^2+4*b*e1*f3+2*A*a*e2^2-4*A*c*e
  0*e2+4*C*a*e0*e2-4*B*a*e1*e2-4*A*b*e1*e2-4*C*b*e2*e3-4*B*a*e0*e3
  +4*A*b*e0*e3-4*A*c*e1*e3-4*C*a*e1*e3-4*B*c*e2*e3+4*B*c*e0*e1-4*C
  *b*e0*e1+4*f1^2+4*f0^2+4*f3^2+4*f2^2:
>
> # Coordinates of base anchor points
> A1:=0:B1:=0:C1:=0:
> A2:=0:B2:=1:C2:=0:
> A3:=0:B3:=-1:C3:=0:
> A4:=1:B4:=0:C4:=0:
> A5:=1:B5:=B:C5:=0:
>
> # Coordinates of platform anchor points
> a1:=0:b1:=0:c1:=0:
> a2:=0:b2:=b2:c2:=0:
> a3:=0:b3:=b3:c3:=0:
> a4:=a:b4:=b4:c4:=0:
> a5:=a:b5:=b5:c5:=0:
>
> # Set of equations
> Lambda1:=factor(subs(a=a1,b=b1,c=c1,A=A1,B=B1,C=C1,R=R1,Lambda))
  :
> Lambda2:=factor(subs(a=a2,b=b2,c=c2,A=A2,B=B2,C=C2,R=R2,Lambda))
  :
> Lambda3:=factor(subs(a=a3,b=b3,c=c3,A=A3,B=B3,C=C3,R=R3,Lambda))
  :
> Lambda4:=factor(subs(a=a4,b=b4,c=c4,A=A4,B=B4,C=C4,R=R4,Lambda))
  :
> Lambda5:=factor(subs(a=a5,b=b5,c=c5,A=A5,B=B5,C=C5,R=R5,Lambda))
  :
> N:=e0^2+e1^2+e2^2+e3^2:
> Phi:=e0*f0+e1*f1+e2*f2+e3*f3:
>
> Delta21:=Lambda2-Lambda1:
```

```
> Delta31:=Lambda3-Lambda1:
> Delta41:=Lambda4-Lambda1:
> Delta51:=Lambda5-Lambda1:
>
> # Item 1
> e3:=solve(a*e0*e1-a*e2*e3-e0*e1-e2*e3,e3);
```

$$\frac{e0\,e1\,(a-1)}{e2\,(a+1)}$$

```
> # (a) Triples m1,m4,m5 and M1,M4,M5 are not congruent
> solve({Phi,Delta41,Delta51},{f1,f2,f3}):
> assign(%);
> G2:=numer(simplify(Delta21)):nops(%);degree(G2,{e0,e1,e2,e3});
```

$$865$$

$$6$$

```
> G3:=numer(simplify(Delta31)):nops(%);degree(G3,{e0,e1,e2,e3});
```

$$865$$

$$6$$

```
> H2:=factor(resultant(G2,N,e0)):nops(%);
```

$$5$$

```
> H3:=factor(resultant(G3,N,e0)):nops(%);
```

$$5$$

```
> K2:=op(1,H2/64/e2^2/a^2/e1^2):
> K3:=op(1,H3/64/e2^2/a^2/e1^2):
>
> factor(coeff(K2,e2,4));
```

$$(a+1)^2\,(b2\,b4-b2\,b5+B)$$

```
> B:=solve(b2*b4-b2*b5+B,B);
```

$$-b2\,b4+b2\,b5$$

```
> factor(coeff(K3,e2,4));
```

$$-(a+1)^2\,(b4-b5)\,(b2+b3)$$

```
> b2:=-b3:
> factor(coeff(K2,e2,2));
```

$$-4\,b3\,e1^2\,(b4-b5)\,(a^2-b3\,b4-b4\,b5-1)$$

```
> factor(coeff(K3,e2,2));
```

$$-4\,b3\,e1^2\,(b4-b5)\,(a^2+b3\,b4-b4\,b5-1)$$

```
> b4:=0:
> factor(coeff(K2,e2,2));
```

$$4\,b3\,b5\,e1^2\,(a-1)\,(a+1)$$

```
> a:=1:
> factor(K2);
```

$$4\,b3\,b5\,e1\,e2^3\,(b5+1)\,(b3+1)$$

```
> factor(K3);
```

$$4\,b3\,b5\,e1\,e2^3\,(b5-1)\,(b3+1)$$

```
> a:='a':b4:='b4':b2:='b2':B:='B':
> f1:='f1':f2:='f2':f3:='f3':
>
> # (b) Triples m1,m4,m5 and M1,M4,M5 are congruent
> a:=1:b4:=0:b5:=B:
> solve({Phi,Delta21,Delta51},{f1,f2,f3}):
> assign(%);
> G4:=numer(simplify(Delta41)):nops(%);degree(G3,{e0,e1,e2,e3});
```

$$14$$
$$6$$

```
> H4:=factor(resultant(G4,N,e0));
```

$$16\,B^2\,e1^2\,e2^2\,(B\,e1-e2)^2$$

```
> a:='a':b4:='b4':b5:='b5':
> f1:='f1':f2:='f2':f3:='f3':e3:='e3':
>
> # Item 2
> e2:=0:
> a:=1:
>
> # (a) Triples m1,m4,m5 and M1,M4,M5 are not reflection-congruent
> solve({Phi,Delta41,Delta51},{f1,f2,f3}):
> assign(%);
> G2:=numer(simplify(Delta21)):nops(%);degree(G2,{e0,e1,e2,e3});
```

$$134$$
$$3$$

```
> G3:=numer(simplify(Delta31)):nops(%);degree(G3,{e0,e1,e2,e3});
```

$$134$$
$$3$$

```
> H2:=factor(resultant(G2,N,e1)):nops(%);
```

$$3$$

```
> H3:=factor(resultant(G3,N,e1)):nops(%);
```

$$3$$

```
> K2:=op(1,H2/16/e0^2):
> K3:=op(1,H3/16/e0^2):
> factor(coeff(K2,e0,2));
```

$$B\,b4\,(b2-b5)$$

```
> factor(coeff(K3,e0,2));
```

$$B\,b4\,(b3-b5)$$

```
>
> # (i)
> b4:=0:
> factor(coeff(K2,e3,2));
```

$$b2\,b5-B$$

```
> B:=b2*b5:
> factor(coeff(K3,e3,2));
```

$$-b5\,(b2+b3)$$

```
> b2:=-b3:
> factor(K2);
```

$$-b3\ b5\ e0\ e3\ (b5-1)\ (b3-1)$$

```
> factor(K3);
```

$$-b3\ b5\ e0\ e3\ (b5+1)\ (b3-1)$$

```
> b4:='b4':b2:='b2':B:='B':
>
> # (ii)
> b2:=b5:b3:=b5:
> factor(coeff(K2,e3,2));
```

$$-b4\ b5+b5^2-B$$

```
> B:=solve(-b4*b5+b5^2-B,B);
```

$$-b4\ b5+b5^2$$

```
> factor(coeff(K3,e3,2));
```

$$2\ b5\ (b4-b5)$$

```
> b3:='b3':b2:='b2':B:='B':
> f1:='f1':f2:='f2':f3:='f3':e3:='e3':
>
> # (b) Triples m1,m4,m5 and M1,M4,M5 are reflection-congruent
> b4:=0:b5:=-B:
> solve({Phi,Delta21,Delta51},{f1,f2,f3}):
> assign(%);
> G4:=numer(simplify(Delta41)):nops(%);degree(G3,{e0,e1,e2,e3});
```

$$14$$

$$3$$

```
> H4:=factor(resultant(G4,N,e0));
```

$$16\ B^2\ e3^2\ (e1^2+e3^2)\ (B^2\ e1^2+B^2\ e3^2+e3^2)$$

```
> # This finishes the discussion of the special case.
>
```